Computer-Assisted Reasoning about Algebraic Topology *

Adam Grabowski[i] and Artur Korniłowicz[ii]

[i]Institute of Mathematics
University of Białystok, Poland
adam@math.uwb.edu.pl

[ii]Institute of Computer Science
University of Białystok, Poland
arturk@math.uwb.edu.pl

Abstract. We describe the motivation and main ideas of the formal approach to the algebraic topology which was developed recently. The work provides some machinery allowing the authors of Mizar articles to obtain more advanced and more valuable results, mainly in general topology, but also in algebra. Some aspects of the Mizar language are also discussed based on the examples taken directly from our developments.

1 Introduction

What is algebraic topology? To answer this question briefly, one could say – it is a part of mathematics in which machinery of abstract algebra is used to analyze properties of topological spaces. The idea of algebraic topology is to transform, in the sense of category theory, a topological space into a group (fundamental, homology or cohomology) and make reasoning inside the group. The key point is that homeomorphic spaces are always transformed into isomorphic groups.

This discipline has been successfully applied in proofs of theorems explicitly involving subsets of n-dimensional Euclidean spaces, like:

- the Brouwer fixed point theorem,
- the Borsuk-Ulam theorem about antipodal points,

but also theorems that, at first glance, are not related to topology, such as:

- fundamental theorem of algebra,
- the Nielsen-Schreier theorem about subgroups of a free group.

The rapid development of topology and abstract algebra in the Mizar system could have been observed from the very beginnings of the Mizar Mathematical Library (MML) dated back to 1989 (or even before as we point out in the next section).

* This work has been partially supported by the FP6 IST project no. 510996 *Types for Proofs and Programs* (TYPES).

ISBN 978-83-7431-128-1 ISSN 0860–150X

As the interconnections between topology and abstract algebra are clearly seen and represented in the literature, moreover both disciplines are rather nicely encoded in the MML, the idea arose to benefit from these earlier developments. On the other hand, we wanted to test if the Mizar language itself has enough expressive power to allow the authors for the reasoning not only in separate topics – like the algebra and the topology, as it was done before, but also if it can be used for exploring theories somewhere on their intersection.

So, the question arose, why not to mechanize another broad area of mathematics? Once set-topology was easy choice (remember the foundations of the MML are provided by axioms of Tarski-Grothendieck set theory), the algebraic topology is located somewhat further from the core of the Mizar system.

Incidentally, group theory and category theory were also represented in the MML (as a matter of fact, any of four subdisciplines under consideration was developed by different people than the remaining ones and the cooperation was rather small).

Although the book of Greenberg [6] was chosen as the background of our work, two main tasks were crucial at early stages of this new formalization project. Originally, as for the time [4] has been written, the primary aim was to provide lemmas for the proof of the Jordan Curve Theorem. As the other, and of minor importance, the development of algebraic topology was pointed out. As it soon appeared, due to some technical time-consuming difficulties within the JCT proof sketch, also the Brouwer fix point theorem[1] could be used to complete the proof. So, the situation changed substantially and since some works towards algebraic topology were yet available in the MML, the directions were as follows:

- first, to prove the Brouwer fix point theorem [7],
- then, using the above, to complete JCT [3].

2 From First Experiments Towards Mechanized Topology

General topology was one of the topics chosen for formalization as a starting point when establishing MML. This choice was by no means accidental – since the Mizar system is based on set theory, just set theory and topology seemed to be a promising stage for early experiments.

Shortly thereafter the fundamentals of the theory were translated into the Mizar language, one of the articles of the more advanced type was included in the library. Its MML identifier was `BORSUK_1` [21] to emphasize the influence of the work of the famous Polish mathematician Karol Borsuk[2].

This article, as [16] claims, was written in 1981 to test Mizar-2 experimental release, much before the regular collecting of Mizar articles was started; it was the translation of the paper *On the homotopy types of some decomposition spaces* by Borsuk into a machine-understandable language. Four years before, a similar effort,

[1] Authors are greatful to Professor Yasunari Shidama of Shinshu University for proposing to formalize the theorem.

[2] Karol Borsuk was advisor of Andrzej Trybulec's PhD in 1974.

but using older Mizar-PC, the formalization of the paper by Krasinkiewicz on the homeomorphism of the Sierpiński curve, was also prepared. These experiments had not much in common with the activities extending the real repository of computer-checked mathematical facts – the article had to have the environment section fully axiomatized, i.e. all preliminary lemmas could be stated without proofs.

At first sight, algebraic topologists use extensively also the apparatus of category theory; even if it is often the case of very simple properties only, and the theorems which are used are not very advanced (especially diagrams are useful here), also this discipline is one of those chosen once as a primary testbed for the Mizar system.

One can recognize two streams of the category theory reflected in the formalization – one of them, authored mainly by Byliński, even though containing a number of results of the general interest, and especially constructions of various category structures as examples, seems not to be developed nowadays. As `CAT_1` was issued in 1989, from that time the language changed significantly, especially the part devoted to the structure handling. The other approach based on [19], which fully exploits system and language capabilities, i.e. categories without uniqueness of the domain and codomain (with identifiers `ALTCAT` standing for alternative categories) was designed and partially implemented by Andrzej Trybulec.

As the example, Cartesian category based on `ProdCatStr` (`CAT` series) has 10 selectors, which is twice as big as the reasonable limit of this parameter; recently the Library Committee of the Association of Mizar Users works on pushing the average number of selectors down. The authors are usually requested not to put too many type information into structures, rather to carry information via attributes; to be more explicit, the unity in the group should not be defined as a selector (especially because we still need an axiom that the unity behaves well), but rather to assure axiomatically that such unity exists.

3 Basic Notions of Formal Algebraic Topology

As one could see in the previous section, algebraic topology merges different theories: topology, group theory and not necessarily explicitly, but using some auxiliary notions, category theory. Of course, for the moment only a small part of algebraic topology has been formalized in Mizar, but large enough to require having developed mentioned theories in the MML. Our Mizar articles containing formalization of the topic are quite clear, in the sense, that we had to prove neither lemmas about pure group theory nor topology.

In spite of the category theory can be explicitly involved in the development of algebraic topology, because for example, fundamental group can be seen as a functor from the category of pointed topological spaces with their homeomorphisms being morphisms in the category into the category of groups with their isomorphisms being morphisms, we decided not to do it.

As we already mentioned, the idea of thorough formalization of algebraic topology evolved from the role of the "side-effect" of JCT project into top priority task. This is reflected in the identifiers of files stored in the MML: while five first belong to the `BORSUK` series (here [4] and [5] will be briefly examined), the remaining, containing more advanced results, form TOPALG series, which will be described in the next section.

3.1 Paths

Let us start with the very basic, although important notion, of a continuous function from the unit interval into a given topological space, which satisfies some additional properties:

```
definition let T be TopStruct; let a, b be Point of T;
  assume a, b are_connected;
  mode Path of a, b -> Function of I[01], T means
:: BORSUK_2:def 2
  it is continuous & it.0 = a & it.1 = b;
end;
```

The assumption allowing to show that an object stated in the definiens exists, is of course needed to prove the correctness of the definition. The definition itself is permissive – it states that if the mapping connecting two points, say a and b, exists, we call it a `Path`, otherwise the definiens is not accessible, even if the Mizar analyser accepts the type `Path of a, b` with no errors reported.

```
definition let T be arcwise_connected TopStruct;
  let a, b be Point of T;
  redefine mode Path of a, b means
:: BORSUK_2:def 4
  it is continuous & it.0 = a & it.1 = b;
end;
```

Although the earlier definition is of a slightly more general type (i.e. it assumes the existence of the mapping connecting only two fixed points), every time that the path returns proper values on its limits should be justified via predicate `are_connected` which seems to be too high price for that minor generalization. Hence if the considered space possesses the property of being `arcwise_connected` (which truly speaking should be rather *pathwise connected*), paths behave as expected (and e.g. real Euclidean line or the unit interval has this adjective added automatically to its type) and the reasoning simplifies as the user can forget about the assumption from the original definition of a path.

3.2 Homotopies

Very similar trick to the aforementioned, i.e. first to guarantee that the desired object exists, then to prove the correctness of the appropriate functor definition, was applied to the definition of a homotopy.

```
definition let T be non empty TopStruct;
  let a, b be Point of T;
```

```
  let P, Q be Path of a, b;
  pred P, Q are_homotopic means
:: BORSUK_2:def 7
  ex f being Function of [:I[01],I[01]:], T st
    f is continuous &
    for s being Point of I[01] holds f.(s,0) = P.s & f.(s,1) = Q.s &
    for t being Point of I[01] holds f.(0,t) = a & f.(1,t) = b;
  symmetry;
end;
```

Under this assumption the existence of a continuous deformation can be proven; of course not its uniqueness, hence the definition of the `mode`, not of the functor.

```
definition let T be non empty TopSpace;
           let a, b be Point of T;
           let P, Q be Path of a, b;
  assume P, Q are_homotopic;
  mode Homotopy of P, Q -> Function of [:I[01],I[01]:], T means
:: BORSUK_6:def 13
  it is continuous &
  for s being Point of I[01] holds it.(s,0) = P.s & it.(s,1) = Q.s &
  for t being Point of I[01] holds it.(0,t) = a & it.(1,t) = b;
end;
```

As among five arguments of this definition two visible are enough, the definition is pretty readable.

4 Fundamental Groups

The relation `are_homotopic` is reflexive, symmetric and transitive, hence all loops of any pointed topological space can be divided into its equivalence classes. Next, it is possible to define a binary operation on the set of all equivalence classes saying that the result class is generated by the sum of representants of its arguments. Such an operation is well-defined and gives the group with the identity being the class generated by the trivial loop at the basepoint and inverses being classes generated by inverses of representants of the classes. The group is called the fundamental group of the space at the basepoint. The formal definition of the group, its fundamental properties, that is the independence (up to the isomorphism) of the fundamental groups from the choice of basepoints from a path-connected component of the space; isomorphism of fundamental groups of homeomorphic topological spaces are presented in next sections. Moreover, examples of fundamental groups of some spaces are listed.

4.1 Definition

The Mizar functor, which makes a group based on a topological space was created in a typical (for Mizar) way, that is, first fields of a group structure (`HGrStr`, [22]) (its carrier and binary operation) were described.

```
definition
  let X be non empty TopSpace, a be Point of X;
  func FundamentalGroup(X,a) -> strict HGrStr means
:: TOPALG_1:def 3
  the carrier of it = Class EqRel (X,a) &
  for x, y being Element of it ex P, Q being Loop of a st
    x = Class(EqRel(X,a),P) & y = Class(EqRel(X,a),Q) &
    (the mult of it).(x,y) = Class(EqRel(X,a),P+Q);
end;
```

Since in the so-called *paper mathematics* fundamental groups are usually denoted as $\pi_1(X, a)$, an appropriate synonym has been introduced.

```
notation
  let X be non empty TopSpace, a be Point of X;
  synonym pi_1(X,a) for FundamentalGroup(X,a);
end;
```

Then, underlying properties (non-emptiness, associativity, the existence of the unity and inverses) of the carrier and the operation expressed as a cluster of adjectives have been proven.

```
registration
  let X be non empty TopSpace;
  let a be Point of X;
  cluster pi_1(X,a) -> non empty associative Group-like;
end;
```

The above mentioned registration ensures that `pi_1(X,a)` is a group. All details are stored in the Mizar article [14].

In the next papers in the series fundamental groups of some basic topological spaces have been computed. Examples are listed in Tab. 1.

Table 1. Examples of fundamental groups of some common topological spaces

space S	$\pi_1(S)$
convex spaces	$\{0\}$
circle	$\mathbb{Z}$
torus	$\mathbb{Z} \times \mathbb{Z}$

4.2 Fundamental Groups of Convex Subspaces of Euclidean Spaces

One of the basic examples of fundamental groups are those computed for convex subspaces of n-dimensional Euclidean spaces. Since all paths between two given points of a convex space are homotopic, the group contains exactly one element – the equivalence class represented by a loop, what is expressed in [9] by the functorial registration:

```
registration
  let n be Element of NAT,
      T be non empty convex SubSpace of TOP-REAL n,
      a be Point of T;
  cluster pi_1(T,a) -> trivial;
end;
```

The homotopy is established by the function:

```
definition
  let n be Element of NAT,
      T being non empty convex SubSpace of TOP-REAL n,
      a, b be Point of T,
      P, Q be Path of a,b;
  func ConvexHomotopy(P,Q) -> Function of [:I[01],I[01]:], T means
:: TOPALG_2:def 2
  for s, t being Element of I[01],
      a1, b1 being Point of TOP-REAL n st a1 = P.s & b1 = Q.s holds
   it.(s,t) = (1-t) * a1 + t * b1;
end;
```

4.3 Fundamental Groups of Homeomorphic Spaces

The important property of fundamental groups is the fact that they are invariants of homeomorphic topological spaces. In [11] we proved the following

```
theorem :: TOPALG_3:35
  for S being non empty TopSpace,
      T being non empty arcwise_connected TopSpace,
      s being Point of S, t being Point of T st
  S,T are_homeomorphic holds pi_1(S,s),pi_1(T,t) are_isomorphic;
```

To get rid of the pathwise connectedness the theorem can be weakened to the form: `pi_1(S,s)` and `pi_1(T,h(s))` are isomorphic when `h` is a homeomorphism from `S` onto `T`. An isomorphism can be established by a function from `pi_1(S,s)` to `pi_1(T,h(s))`, which assigns the equivalence class represented by the loop `h*f` at the point `h(s)` of `T`, where `f` is a loop at a given point `s` of `S`.

4.4 Fundamental Groups of Simple Closed Curves

In many papers devoted to algebraic topology the fundamental group of a circle, which is isomorphic to the group of integers, is mentioned. Having results presented in Sec. 4.3 it is easy to generalize the theorem to any simple closed curve, which in fact we did in [12]:

```
theorem :: TOPALG_5:27
  for S being being_simple_closed_curve SubSpace of TOP-REAL 2,
      x being Point of S holds
   INT.Group, pi_1(S,x) are_isomorphic;
```

The proof of the isomorphism for any simple closed curve can be started from an easier case. It is enough to compute the fundamental group of the unit circle centered at the point $(0, 0)$ based at the point $(1, 0)$, and applying the independence of the fundamental group from the base point pass to the fundamental group of the unit circle centered at the point $(0, 0)$ based at any point, and finally using isomorphism of fundamental groups of homeomorphic spaces pass from the unit circle to the general case, that is to simple closed curves.

4.5 Fundamental Groups of Products of Topological Spaces

Another property establishes correspondence between fundamental groups and the product of topological spaces. The main result of [10] states that the fundamental group of the product of two topological spaces is isomorphic to the product of the fundamental groups of the spaces.

```
theorem :: TOPALG_4:32
  for S, T being non empty arcwise_connected TopSpace,
      s1, s2 being Point of S,
      t1, t2 being Point of T holds
  pi_1([:S,T:],[s1,t1]), product <*pi_1(S,s2),pi_1(T,t2)*>
                                   are_isomorphic;
```

The isomorphism is established by the function:

```
definition
  let S, T be non empty TopSpace,
      s be Point of S, t be Point of T;
  func FGPrIso(s,t) ->
    Function of pi_1([:S,T:],[s,t]), product <*pi_1(S,s),pi_1(T,t)*>
      means
:: TOPALG_4:def 2
  for x being Point of pi_1([:S,T:],[s,t])
   ex l being Loop of [s,t] st x = Class(EqRel([:S,T:],[s,t]),l) &
   it.x = <*Class(EqRel(S,s),pr1 l),Class(EqRel(T,t),pr2 l)*>;
end;
```

One trivial consequence of the theorem is that the fundamental group of the torus is isomorphic to $\mathbb{Z} \times \mathbb{Z}$ – the torus can be represented as the product of two circles, which are simple closed curves.

4.6 The Brouwer Fixed Point Theorem

The main result of developing of algebraic topology in Mizar is the formalization of the Brouwer fixed point theorem for 2-dimensional Euclidean spaces, see [13].

```
theorem :: BROUWER:14
  for r being non negative (real number),
      o being Point of TOP-REAL 2,
      f being continuous Function of Tdisk(o,r), Tdisk(o,r) holds
  f has_a_fixpoint;
```

Its proof (taken directly from [7]) is mainly based on the fact that a circle is not a retract of a disk; and this is the place where machinery of algebraic topology is used. What is needed here is the fact that fundamental group of a circle is non trivial. This is a simple consequence of the fact that fundamental group of a circle, which is a simple closed curve, is isomorphic to the group of integers, which is infinite, that is non-trivial.

5 Conclusions

Although the mechanization of the algebraic topology into Mizar formalism is still at very early stage (some ten articles devoted to the topic out of 960 in the MML), we can estimate the influence of this effort for the system and the library very positively. The development of the fundamentals helped us to speed up the formalization of the proof of the Jordan Curve Theorem in Mizar, and the Brouwer fix point theorem [13] was completed also employing algebraic topology techniques. Another large project of the Mizar team, the translation of the *Compendium of Continuous Lattices*, is still ongoing, although currently at a much slower pace than at the beginning. In this case structures of topological spaces were successfully merged with posets.

There are also articles forming `TOPGEN` series which cover [15] but this work just started. However, in parallel to broad formalization efforts, within the Mizar Mathematical Library also problems which are more a kind of logical puzzles can be solved – as fourteen Kuratowski's sets or complete formalization of a chosen paper – [8].

Acknowledgements

The authors would like to express their gratitude to Andrzej Trybulec for acquainting them with Mizar and his continuous support since authors‘ students times, as well as for turning our attention to the formalization of topology.

A Mizar Notations Used in the Paper

× [:X,Y:] is the Cartesian product of sets X and Y ([1])
× I[01] is the interval [0,1] with standard topology ([21])
× INT.Group is the additive group of integers ([20])
× HGrStr is a structure of a group (equivalent of magma) ([22])
× TopStruct is a structure of a topological space ([17])
× TopSpace is a topological space ([17])
× TOP-REAL n is an n-dimensional Euclidean space ([2])
× EqRel is the relation being_homotopic between loops at a given point ([14])
× Class is the set of all equivalence classes of a given equivalence relation ([18])

References

1. Czesław Byliński. Some basic properties of sets. *Formalized Mathematics*, 1(**1**):47–53, 1990. MML Id: ZFMISC_1.
2. Agata Darmochwał. The Euclidean space. *Formalized Mathematics*, 2(**4**):599–603, 1991. MML Id: EUCLID.
3. David Gauld. Brouwer's Fixed Point Theorem and the Jordan Curve Theorem. http://aitken.math.auckland.ac.nz/~gauld/750-05/section5.pdf.
4. Adam Grabowski. Introduction to the homotopy theory. *Formalized Mathematics*, 6(**4**):449–454, 1997. MML Id: BORSUK_2.
5. Adam Grabowski and Artur Korniłowicz. Algebraic properties of homotopies. *Formalized Mathematics*, 12(**3**):251–260, 2004. MML Id: BORSUK_6.
6. Marvin J. Greenberg. *Lectures on Algebraic Topology.* W. A. Benjamin, Inc., 1973.
7. Allen Hatcher. *Algebraic Topology.* Cambridge University Press, 2002.
8. Yoshinori Isomichi. New concepts in the theory of topological space – supercondensed set, subcondensed set, and condensed set, *Pacific Journal of Mathematics,* 38(**3**):657–668, 1971.
9. Artur Korniłowicz. The fundamental group of convex subspaces of $\mathcal{E}^n_{\mathrm{T}}$. *Formalized Mathematics*, 12(**3**):295–299, 2004. MML Id: TOPALG_2.
10. Artur Korniłowicz. On the fundamental groups of products of topological spaces. *Formalized Mathematics*, 12(**3**):421–425, 2004. MML Id: TOPALG_4.
11. Artur Korniłowicz. On the isomorphism of fundamental groups. *Formalized Mathematics*, 12(**3**):391–396, 2004. MML Id: TOPALG_3.
12. Artur Korniłowicz. The fundamental group of the circle. *Formalized Mathematics*, 13(**2**):325–331, 2005. MML Id: TOPALG_5.
13. Artur Korniłowicz and Yasunari Shidama. Brouwer fixed point theorem for disks on the plane. *Formalized Mathematics*, 13(**2**):333–336, 2005. MML Id: BROUWER.
14. Artur Korniłowicz, Yasunari Shidama, and Adam Grabowski. The fundamental group. *Formalized Mathematics*, 12(**3**):261–268, 2004. MML Id: TOPALG_1.
15. Kazimierz Kuratowski. *Topology*, volume II. PWN – Polish Scientific Publishers, Academic Press, Warsaw, New York and London, 1968.
16. Roman Matuszewski and Piotr Rudnicki. Mizar: the first 30 years, *Mechanized Mathematics and Its Applications,* 4(**1**):3–24, 2005, http://mizar.org/people/romat/MatRud2005.pdf.
17. Beata Padlewska and Agata Darmochwał. Topological spaces and continuous functions. *Formalized Mathematics*, 1(**1**):223–230, 1990. MML Id: PRE_TOPC.

18. Konrad Raczkowski and Paweł Sadowski. Equivalence relations and classes of abstraction. *Formalized Mathematics*, 1(**3**):441–444, 1990. MML Id: `EQREL_1`.
19. Zbigniew Semadeni and Antoni Wiweger. *Wstęp do teorii kategorii i funktorów*, volume 45 of *Biblioteka Matematyczna*. PWN, Warszawa, 1978.
20. Dariusz Surowik. Cyclic groups and some of their properties – part I. *Formalized Mathematics*, 2(**5**):623–627, 1991. MML Id: `GR_CY_1`.
21. Andrzej Trybulec. A Borsuk theorem on homotopy types, *Formalized Mathematics*, 2(**4**):535–545, 1991. MML Id: `BORSUK_1`.
22. Wojciech A. Trybulec. Groups. *Formalized Mathematics*, 1(**5**):821–827, 1990. MML Id: `GROUP_1`.